Parciální derivace

Základní údaje

Parciální derivace se počítají stejně jako derivace funkce jedné proměnné, jen nelze používat čárku jako horní index pro označení derivace. Proměnné v příkazu D po uvedení funkce se píší v pořadí, v jakém se mají derivovat. Pokud na pořadí nezáleží, lze jednotlivé proměnné seskupit dohromady, y,x,y lze psát jako x,{y,2}. Hodnoty získáme dosazením x->.., y->.. .

```
f[x_, y_] := x^5 - y^3 + x^2 y^4
```

```
D[f[x, y], x, y]
```

$8\,x\,y^3$

```
D[f[x, y], y, x, y]
```

$24\,x\,y^2$

```
D[f[x, y], x, {y, 2}]
```

$24\,x\,y^2$

```
D[f[x, y], x, {y, 2}] /. {x → 2, y → 3}
```

432

V případě, že na pořadí nezáleží, lze seskupení proměnných psát i následovně. V tomto případě můžeme i přímo počítat číselné hodnoty parciálních derivací. Pokud nezadáme proměnné, je výsledkem ,,pure” funkce

```
Derivative[1, 2][f][x, y]
```

$24\,x\,y^2$

```
Derivative[1, 2][f][2, 3]
```

432

```
Derivative[1, 2][f]
```

$24 \#1 \#2^2 \&$

Některé proměnné mohou záviset na jiných. To se vyjádří volbou NonConstants->{y,...}. Nejdříve se podíváme na funkci dvou proměnných, kde y je funkcí x (tj. jedná se o implicitní funkci). Ve výsledku můžeme místo D[y,x,NonConstants→{y}] psát y'. V dalším příkladě funkce tří proměnných je D[z,x,NonConstants→{z}] parciální derivace z podle x, atd.

```
D[f[x, y], x, NonConstants → y]
```

$4 x^2 y^3 D[y, x, \text{NonConstants} \to \{y\}] - 3 y^2 D[y, x, \text{NonConstants} \to \{y\}] + 5 x^4 + 2 x y^4$

```
g[x_, y_, z_] := x^2 y^3 - z^2 y^4 + 5 z^2 x^7
```

```
D[g[x, y, z], x, NonConstants → z]
```

$10 x^7 z D[z, x, \text{NonConstants} \to \{z\}] - 2 y^4 z D[z, x, \text{NonConstants} \to \{z\}] + 35 x^6 z^2 + 2 x y^3$

```
D[g[x, y, z], x, y, NonConstants → z]
```

$10 x^7 D[z, x, \text{NonConstants} \to \{z\}] D[z, y, \text{NonConstants} \to \{z\}] + 10 x^7 z D[z, x, y, \text{NonConstants} \to \{z\}] + 70 x^6 z D[z, y, \text{NonConstants} \to \{z\}] - 2 y^4 D[z, x, \text{NonConstants} \to \{z\}] D[z, y, \text{NonConstants} \to \{z\}] - 2 y^4 z D[z, x, y, \text{NonConstants} \to \{z\}] - 8 y^3 z D[z, x, \text{NonConstants} \to \{z\}] + 6 x y^2$

Nespojitá funkce mající všechny směrové derivace v daném bodě.

Následující příklad ukazuje nespojitou funkci v počátku, která v něm má obě parciální derivace rovny 0, Nakreslíme graf funkce s tečnaqmi v počátku rovnoběžnými s osami souřadnic.

```
Clear[f, fx, fy]; f[x_, y_] := x^2 y / (x^4 + y^2)
```

```
fx[0, 0] = Limit[f[h, 0] / h, h → 0]
fy[0, 0] = Limit[f[0, h] / h, h → 0]
```

0

0

Tato funkce má v počátku dervace dokonce v libovolném směru (u,v).

```
Limit[f[h u, h v] / h, h → 0]
```

$$\frac{u^2}{v}$$

```
p1 = Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1},
   PlotStyle -> Opacity[.5]];
p2 = ParametricPlot3D[{x, 0, 0}, {x, -1, 1},
   PlotStyle → {Thickness[.02], Red}];
p3 = ParametricPlot3D[{0, x, 0}, {x, -1, 1},
   PlotStyle → {Thickness[.02], Green}];
Show[p1, p2, p3]
```



Nezáměnnost smíšených derivací

Můžeme vizualizovat známý příklad na nezáměnnost smíšených parciálních derivací.

```
Clear[f]; f[x_, y_] := x * y * (x^2 - y^2) / (x^2 + y^2)
```

```
Plot3D[f[x, y], {x, -3, 3}, {y, -3, 3}]
```



Vypočteme parciální derivaci podle x pro x≠0,y=0 a potom podle definice v bodě (0,0).

```
fx[x_, y_] = D[f[x, y], x]
```

$$\frac{2 x^2 y}{x^2 + y^2} - \frac{2 x^2 y \left(x^2 - y^2\right)}{\left(x^2 + y^2\right)^2} + \frac{y \left(x^2 - y^2\right)}{x^2 + y^2}$$

```
fx[0, 0] = Limit[f[h, 0] / h, h → 0]
```

$$0$$

Podobně vypočteme parciální derivaqci podle y.

```
fy[x_, y_] = D[f[x, y], y]
```

$$-\frac{2\,x\,y^2}{x^2+y^2}-\frac{2\,x\,y^2\left(x^2-y^2\right)}{\left(x^2+y^2\right)^2}+\frac{x\left(x^2-y^2\right)}{x^2+y^2}$$

```
fy[0, 0] = Limit[f[0, h] / h, h → 0]
```

$$0$$

Nyní spočteme parciální derivaci podle y v (0,0) z parciální derivace podle x a potom parciální derivaci podle x v (0,0) z parciální derivace podle y. Vidíme, že hodnoty jsou různé. Plochy obou parciálních derivací funkce podle x a podle y je vidět na spodních grafech (snadno vidíte, že směrnice tečen v (0,0) na obou plochách podle os jsou různé).

```
fxy = Limit[fx[0, h] / h, h → 0]
```

$$-1$$

```
fyx = Limit[fy[h, 0] / h, h → 0]
```

$$1$$

```
{Plot3D[fx[x, y], {x, -3, 3}, {y, -3, 3}, AxesLabel → {x, y, z},
   ImageSize → 240],
 Plot3D[fy[x, y], {x, -3, 3}, {y, -3, 3}, AxesLabel → {x, y, z},
   ImageSize → 240]}
```



Parciální derivace složené funkce

Můžeme si připomenout parciální derivace složené funkce. *Mathematica* umí přímo spočítat derivaci složené funkce jen po dosazení vnitřních funkcí ve vypočítaném tvaru pro dosazované proměnné. To je samozřejmě jednoduché.

```
Clear[f, g, h]; D[f[g[u, v], h[u, v]], u]
D[f[g[u, v], h[u, v]], u, v]
```

$$g^{(1,0)}(u,v)\, f^{(1,0)}(g(u,v), h(u,v)) + h^{(1,0)}(u,v)\, f^{(0,1)}(g(u,v), h(u,v))$$

$$h^{(1,0)}(u,v)\left(g^{(0,1)}(u,v)\, f^{(1,1)}(g(u,v), h(u,v)) + h^{(0,1)}(u,v)\, f^{(0,2)}(g(u,v), h(u,v))\right) + g^{(1,0)}(u,v)\left(g^{(0,1)}(u,v)\, f^{(2,0)}(g(u,v), h(u,v)) + h^{(0,1)}(u,v)\, f^{(1,1)}(g(u,v), h(u,v))\right) + g^{(1,1)}(u,v)\, f^{(1,0)}(g(u,v), h(u,v)) + h^{(1,1)}(u,v)\, f^{(0,1)}(g(u,v), h(u,v))$$

```
Clear[f, g, h, u, v]; f[x_, y_] := Sin[x^2 y];
g[u_, v_] := u^2 - v^2;
h[u_, v_] := u v;
```

```
D[f[g[u, v], h[u, v]], u, u] + D[f[g[u, v], h[u, v]], v, v]
```

$$\left(u\left(u^2-v^2\right)^2 - 4\,u\,v^2\left(u^2-v^2\right)\right)^2\left(-\sin\left(u\,v\left(u^2-v^2\right)^2\right)\right) - \left(4\,u^2\,v\left(u^2-v^2\right) + v\left(u^2-v^2\right)^2\right)^2 \sin\left(u\,v\left(u^2-v^2\right)^2\right) + \left(8\,u\,v^3 - 12\,u\,v\left(u^2-v^2\right)\right)\cos\left(u\,v\left(u^2-v^2\right)^2\right) + \left(8\,u^3\,v + 12\,u\,v\left(u^2-v^2\right)\right)\cos\left(u\,v\left(u^2-v^2\right)^2\right)$$

```
FullSimplify[%]
```

$$-\left(u^2+v^2\right)\left(\left(u^2-v^2\right)^2\left(u^4+14\,u^2\,v^2+v^4\right)\sin\left(u\,v\left(u^2-v^2\right)^2\right) - 8\,u\,v\cos\left(u\,v\left(u^2-v^2\right)^2\right)\right)$$

Vyjádříme nyní Laplaceův operátor v polárních souřadnicích.

```
Clear[f, g, h]; g[x_, y_] := Sqrt[x^2 + y^2];
h[x_, y_] := ArcTan[y / x];
D[f[g[x, y], h[x, y]], x, x] + D[f[g[x, y], h[x, y]], y, y]
```

$$-\frac{y^2 f^{(1,0)}\left(\sqrt{x^2+y^2}\,,\,\tan^{-1}\left(\frac{y}{x}\right)\right)}{\left(x^2+y^2\right)^{3/2}}-$$

$$\frac{y\left(\frac{x\,f^{(1,1)}\left(\sqrt{x^2+y^2}\,,\tan^{-1}\left(\frac{y}{x}\right)\right)}{\sqrt{x^2+y^2}}-\frac{y\,f^{(0,2)}\left(\sqrt{x^2+y^2}\,,\tan^{-1}\left(\frac{y}{x}\right)\right)}{x^2\left(\frac{y^2}{x^2}+1\right)}\right)}{x^2\left(\frac{y^2}{x^2}+1\right)}+$$

$$\frac{y\left(\frac{f^{(1,1)}\left(\sqrt{x^2+y^2}\,,\tan^{-1}\left(\frac{y}{x}\right)\right)}{x\left(\frac{y^2}{x^2}+1\right)}+\frac{y\,f^{(2,0)}\left(\sqrt{x^2+y^2}\,,\tan^{-1}\left(\frac{y}{x}\right)\right)}{\sqrt{x^2+y^2}}\right)}{\sqrt{x^2+y^2}}+$$

$$\frac{2\,f^{(1,0)}\left(\sqrt{x^2+y^2}\,,\,\tan^{-1}\left(\frac{y}{x}\right)\right)}{\sqrt{x^2+y^2}}-\frac{x^2\,f^{(1,0)}\left(\sqrt{x^2+y^2}\,,\,\tan^{-1}\left(\frac{y}{x}\right)\right)}{\left(x^2+y^2\right)^{3/2}}+$$

$$\frac{\frac{f^{(0,2)}\left(\sqrt{x^2+y^2}\,,\tan^{-1}\left(\frac{y}{x}\right)\right)}{x\left(\frac{y^2}{x^2}+1\right)}+\frac{y\,f^{(1,1)}\left(\sqrt{x^2+y^2}\,,\tan^{-1}\left(\frac{y}{x}\right)\right)}{\sqrt{x^2+y^2}}}{x\left(\frac{y^2}{x^2}+1\right)}+$$

$$\frac{x\left(\frac{x\,f^{(2,0)}\left(\sqrt{x^2+y^2}\,,\tan^{-1}\left(\frac{y}{x}\right)\right)}{\sqrt{x^2+y^2}}-\frac{y\,f^{(1,1)}\left(\sqrt{x^2+y^2}\,,\tan^{-1}\left(\frac{y}{x}\right)\right)}{x^2\left(\frac{y^2}{x^2}+1\right)}\right)}{\sqrt{x^2+y^2}}-$$

$$\frac{2\,y^3\,f^{(0,1)}\left(\sqrt{x^2+y^2}\,,\,\tan^{-1}\left(\frac{y}{x}\right)\right)}{x^5\left(\frac{y^2}{x^2}+1\right)^2}+$$

$$\frac{2\,y\,f^{(0,1)}\left(\sqrt{x^2+y^2}\,,\,\tan^{-1}\left(\frac{y}{x}\right)\right)}{x^3\left(\frac{y^2}{x^2}+1\right)}-\frac{2\,y\,f^{(0,1)}\left(\sqrt{x^2+y^2}\,,\,\tan^{-1}\left(\frac{y}{x}\right)\right)}{x^3\left(\frac{y^2}{x^2}+1\right)^2}$$

```
FullSimplify[%]
```

$$\frac{f^{(0,2)}\left(\sqrt{x^2+y^2},\tan^{-1}\left(\frac{y}{x}\right)\right)}{x^2+y^2}+\frac{f^{(1,0)}\left(\sqrt{x^2+y^2},\tan^{-1}\left(\frac{y}{x}\right)\right)}{\sqrt{x^2+y^2}}+f^{(2,0)}\left(\sqrt{x^2+y^2},\tan^{-1}\left(\frac{y}{x}\right)\right)$$

```
% /. {x^2 + y^2 → r^2, ArcTan[y / x] → t, Sqrt[r^2] → r}
```

$$\frac{f^{(0,2)}\left(\sqrt{r^2},t\right)}{r^2}+\frac{f^{(1,0)}\left(\sqrt{r^2},t\right)}{\sqrt{r^2}}+f^{(2,0)}\left(\sqrt{r^2},t\right)$$